
昨今オブジェクト指向は，さまざまな言語のさまざ

まな分野で活発に研究されています．オブジェクト指

向の概念はプログラミング言語を横断して共有できる

ため，研究結果を共通の資産として活用することがで

きます．つまりオブジェクト指向機能を備えたPerlに

もその研究結果，知識，ノウハウを流用することがで

きるのです．

ところが，CGI用スクリプトとして流通するコード
注1で参考にできるものが少ない上，手続き指向言語と

してのノウハウの蓄積が多いためか，Perlのオブジェ

クト指向機能はあまり積極的に利用されていないのが

現実です．オブジェクト指向がもたらすデータのカプ

セル化，ポリモーフィズム，継承などによるメリット

を無視するのはあまりにもったいないことです．

今回はPerlのオブジェクト指向機能の解説からはじ

まり，Webアプリケーション開発中に発生する問題に

対してデザインパターンを適用して解決してゆく方法

を述べていきます．

オブジェクト指向の構文自体は単純で，誰でもすぐ

にコードを記述することができます．しかしオブジェ

クト指向による設計，それも変更に強く再利用可能な

設計は正直難しいことです．

ではどのようにすれば巧い設計ができるようになる

のでしょうか．

1つは経験を積むことです．もう1つは先人の設計，

特に有用性が認められ幅広く利用されている設計を真

似ることです．経験は時間をかけて積むとして，どの

ような手段でその巧い設計を探し出せば良いのでしょ

うか．その解はデザインパターンを紐解くことです．

デザインパターンはオブジェクト指向のソフトウェ

アを設計する際の定石で，この定石を集めたパターン

カタログは書籍やオンラインで参照することができま

す．デザインパターンはオブジェクト指向のソフトウ

ェアで繰り返し利用される重要な設計に，体系的に名

前をつけ，説明・評価し，その設計経験を第三者が効

果的に利用できるように整理したもの，つまりお約束

集・ネタ帳と言えます．

デザインパターンのパターンカタログには

¡名前：そのパターンの名称

¡目的：そのパターンの主な目的

¡問題：どのような問題に当てはめられるか

¡解法：そのようにその問題を解決するか

¡結果：その設計によってどのようなメリットとデメ

リットが生まれるか

3)デザインパターンを学ぶ
オブジェクト指向と頻出パターンの紹介
㈱クレイフィッシュ 小山 浩之 OYAMA Hiroyuki   oyama@crayfish.co.jp

▼第3章

パターンカタログ

注1）Perl普及の原動力の1つであったとも言えますが．

はじめに

オブジェクト指向の難しさ

デザインパターンとは

がまとめられています．ある程度複雑さをもったもの

を言い表すときに名前が決まっていると，意思疎通や

記憶がスムーズにいきます．自分以外の人にどんな設

計を採用しているか，どんな設計をすれば良いかを伝

えるとき「ここは～を使っているよ」「そこは～にす

れば良くない？」と一言で言えたら便利ですよね．

デザインパターンはErich Gamma，Richard Helm，

Ralph Johnson，John Vlissidesら4人による通称GoF

（Gang Of Four）パターンが有名です．

パターンはこのほかにもさまざまな種類がまとめら

れています．ビジネスモデリングのパターンをまとめ

たアナリシスパターンなどマクロ的な視点のパターン

もあれば，GoFパターンなどのようにミクロ的な視点

のパターンもあります．

◇　◇　◇

今回はさまざまなパターンの中からオブジェクト指

向プログラミングのスキルアップや効率化に有効と思

われる，GoFパターンを主に扱います．

デザインパターンの話に移る前に，まずPerlのオブ

ジェクト指向機能のおさらいをします．

オブジェクトは，あるデータ構造とその処理方法を

持っています．要は変数とそれを操作するための関数

を集めたものとも言えます．オブジェクトを利用する

ためには以下のような手続きを踏みます．

qクラスを定義する

wクラスをインタンス化する

eインスタンスを操作する

rインスタンスを破棄する

それでは日付のデータを日付の文字列に変換する

DateStringクラスを例にして，オブジェクトの利用手

順と用語を確認していきます．

リスト1に挙げたDateStringクラスはDateString.pm

というファイル名で保存します．DateStringクラスを

利用するスクリプト（リスト2）を実行すると

Sun Oct 21 16:59:24 2001

といった現在の日付を表す文字列が出力されます．

■クラスを定義する
クラスの定義はpackageオペレータを使用します．

packageを宣言してからコードの終端に達するか別の

packageを宣言する，もしくはスコープを外れるまで

が1つのクラスを表します．この例では

package DateString;

でDateStringクラスの定義を始めることを宣言してい

ます．

クラスを利用するためにはクラスのインスタンスを

得る必要があり，そのインスタンスを生成するための

メソッドをコンストラクタと呼びます．

コンストラクタはクラスに結び付けたデータのリフ

ァレンスを返すメソッドで，DateStringクラスではサ

ブルーチンnew( )がコンストラクタです．コンストラ

クタnewは，引数で受け取ったdelta秒を格納するハ

パターンの種類

オブジェクト

Perlのオブジェクト
指向のおさらい

●リスト1 DateStringクラス

package DateString;
use strict;

sub new
{

my $class = shift;
my $time = shift || time();
return bless { time => $time }, $class;

}

sub to_string
{

my $self = shift;
return scalar localtime $self->{time};

}
1;
__END__

●リスト2 DateStringクラスの使用

#!/usr/bin/perl
use DateString;

my $date = DateString->new;
print $date->to_string;
__END__

ッシュを作成しそのリファレンスをクラスに結び付け

て返します．引数が指定されない場合はデフォルトの

動作として現在時刻を格納します．

これを少し冗長に記述すると

sub new

{

my $class = shift;

my $time = shift || time();

my $date = { time => $time };

return bless($data, $class);

}

といった記述になります．

また，Perlのコンストラクタには任意の名前をつけ

ることができます．DateStringクラスの例ではnewと

いう一般的な名称を使用していますが，instanceなど

別の名前でもかまいません．

コンストラクタの第1引数には自動的にクラス名が

渡されます．よってnewメソッドの変数$classはクラ

ス名の文字列‘DateString’を受け取っています．コ

ンストラクタに引数を指定する場合は

my $tomorrow = time() + 60 * 60 * 24;

my $date = DateString->new($tomorrow);

と指定します．コンストラクタの内部では第1引数に

クラス名が渡され，指定した引数は第2引数以降に渡

されます（図1）．

データとクラスの結び付けは，blessオペレータが

行います．blessは第1引数にデータのリファレンス，

第2引数にそのデータを結び付けるクラスの名前を指

定します．blessされるデータはハッシュでも配列で

もスカラでも型グロブでも，リファレンスであれば何

でも構いません．

JavaやC++などデフォルトコンストラクタが用意さ

れる言語とは異なり，Perlは明示的にコンストラクタ

を定義する必要があります．これはオブジェクトが内

包するデータに任意のデータ型を使用でき，これを指

定するためです．

◇　◇　◇

クラスをインスタンス化しただけでは何も仕事がで

きませんので，コンストラクタに続いて日付のデータ

をctime形式の文字列に変換するto_stringメソッドを

定義します．

sub to_string

{

my $self = shift;

return scalar localtime $self->{time};

}

メソッドの第1引数は自動的にそのオブジェクトの

インスタンスが渡されます（図2）．この例では変数

$selfにインスタンスを受け取っています．このインス

タンスを通してコンストラクタで定義したデータ構造

にアクセスできます．DateStringクラスで使用するデ

ータはハッシュなので，ハッシュのリファレンスとし

てアクセスします．

$self->{time};

仮にオブジェクトにデータをセットする場合は

my $yesterday = time() - 60 * 60 * 24;

$self->{time} = $yesterday;

package DateString;�
sub new�
a�
 my $class = shift;�
 my $time = shift‖time();�
return bless a time => $tme s, $class;�
s

DateString->new(time()＋06*06*24);

図1／コンストラクタの呼び出し

$date->to_string();

sub to_string�
a�
 my $self = shift;�
 return scalar localtime $self->atimes;�
s

図2／メソッドの呼び出し

といった記述でデータをセットできます．

■クラスをインスタンス化する
これでクラスの定義が終わりました．続いてこのク

ラスを使用してみます．クラスをインスタンス化する

ためにはそのクラスのコンストラクタを呼び出します．

my $date = DateString->new;

この記述で先程定義したコンストラクタnewが実行

され，DateStringクラスのインスタンス（オブジェク

ト）を返します．インスタンスは任意の変数型で保持

することができ，この例ではスカラ変数$dateにイン

スタンスを受け取っています．

■インスタンスを操作する
生成したインスタンスを操作するためにメソッドを

呼び出します．

print $date->to_string;

これによりDateStringクラスで定義したメソッド

to_stringが実行され，変換された日付の文字列が返

されるので，標準出力に

Sun Oct 21 16:59:24 2001

と出力します．

■インスタンスを破棄する
使用されなくなったインスタンスは自動的に破棄さ

れるので，プログラマが明示的に行う必要はありませ

ん．Perlは誰からも参照されなくなった変数（インス

タンス）をガーベッジコレクションの対象として破棄

します．つまりレキシカル変数はスコープを外れるな

ど，他のオブジェクトから参照されなくなった時点で

ガーベッジコレクションの対象になります．

また，インスタンスが消滅するタイミングで呼び出

される特別なメソッド（デストラクタ）を定義するこ

ともできます．DESTROYという名前で定義したメソ

ッドがデストラクタになり，インスタンスが消滅する

直前に呼び出されます．

継承とは，既存のクラスを元にして，機能を追加し

た新しいクラスを少ない手数で定義できる方法です．

また，継承は重複したコードを抜き出して1つにまと

める手段としても利用されます．

継承したクラスは継承する元のクラス（スーパクラ

ス）が実装しているメソッドを引き継ぎます．その引

き継いだメソッドをオーバライド（上書き）して動作

をカスタマイズしたり新しくメソッドを追加したりと

いった，差分だけをコーディングすることが可能にな

ります．これにより既存のコードの編集や新たにコピ

ー&ペーストするような，リスクを伴う変更を行わず

に済みます．

◇　◇　◇

では先ほどのDateStringクラスを継承して，HTTP

で利用されるRFC1123形式の日付の文字列を返す

継承

●リスト3 DateStringRfc1123クラス

package DateStringRfc1123;
use base 'DateString';

use constant RFC1123_FORMAT => '%s, %02d %s %04d %02d:%02d:%02d GMT';

sub to_string
{

my $self = shift;
my @gmt = gmtime $self->{time};
return sprintf RFC1123_FORMAT,

(qw(Sun Mon Tue Wed Thu Fri Sat))[$gmt[6]],
$gmt[3],
(qw(Jan Feb Mar Apr May Jun Jul Aug Sep Oct Nov Dec))[$gmt[4]],
$gmt[5] + 1900,
$gmt[2], $gmt[1], $gmt[0];

}
1;
__END__

DateStringRfc1123クラスを作成します．リスト3を

見てください．

Perlでクラスを継承する場合はbaseプラグマを使用

します．DateStringRfc1123クラスはDateStringクラ

スを継承するために以下のように宣言します．

package DateStringRfc1123;

use base 'DateString';

これによりDateStringRfc1123はDateStringクラス

を継承したクラスとして宣言されます．version 5.004

などの若干古いバージョンのperlを使用する場合は，

baseプラグマを使わずに配列@ISAに継承するクラス

名をセットして指定します．

package DateStringRfc1123;

@ISA = qw(DateString);

この方法は最新のversion 5.6.1などでも使用するこ

とができますが，見通しの良さからbaseプラグマを使

用する方法をお勧めします．

仮に複数のクラスから継承（多重継承）する場合

は，baseプラグマに配列でスーパクラスの名前を指定

します．

use base qw(DateString JapaneseString);

配列@ISAで指定する場合も同様です．

DateStringクラスを継承したDateStringRfc1123ク

ラスでは，to_stringメソッドだけが実装されています

が，DateStringから受け継いだコンストラクタnewを

同様に使用することができます．

リスト4のコードを実行するとRFC1123の形式で現

在の日付の文字列を得ることができます．

Sun, 21 Oct 2001 16:59:24 GMT

DateStringクラスで定義したto_stringメソッドでは

なく，DateStringRfc1123クラスで定義したto_string

メソッドが呼び出されていることがわかります．この

ようにスーパクラスで定義されているメソッドを上書

きしてカスタマイズすることをオーバライドと呼びま

す．

同様にDateStringクラスを継承してRFC850の形式

の日付文字列を出力するDateStringRfc850クラスを定

義しましょう（リスト5）．このクラスで定義した

to_stringメソッドは

Sunday, 21-Oct-01 16:59:24 GMT

という形式で日付の文字列を出力します．

オブジェクト指向プログラミングで得られる最も大

きなメリットの1つとして，動的束縛とポリモフィズ

デザインパターンを学ぶ
オブジェクト指向と頻出パターンの紹介第　　章3

●リスト4 DateStringRfc1123クラスの使用

#!/usr/bin/perl
use DateStringRfc1123;

my $date = DateStringRfc1123->new;
print $date->to_string;
__END__

●リスト5 DateStringRfc850クラス

package DateStringRfc850;
use base 'DateString';

use constant RFC850_FORMAT => '%s, %02d-%s-%02d %02d:%02d:%02d GMT';

sub to_string
{

my $self = shift;
my @gmt = gmtime $self->{time};
return sprintf RFC850_FORMAT,

(qw(Sunday Monday tuesday Wednesday Thursday Friday Saturday))[$gmt[6]],
$gmt[3],
(qw(Jan Feb Mar Apr May Jun Jul Aug Sep Oct Nov Dec))[$gmt[4]],
$gmt[5],
$gmt[2], $gmt[1], $gmt[0];

}
1;
__END__

動的束縛とポリモフィズム

ムを利用できる点が挙げられます．例としてリスト6

を見てください．

このコードはDateString，DateStringRfc1123，Date

StringRfc850それぞれのインスタンスを配列@dateに

格納した後，配列@dateからインスタンスを取り出し

て，無造作にto_stringメソッドを呼び出します．こ

れを実行すると次のような出力が得られます．

Sun Oct 21 16:59:24 2001

Sun, 21 Oct 2001 16:59:24 GMT

Sunday, 21-Oct-01 16:59:24 GMT

もしこのような処理をオブジェクト指向ではない別

の記述の仕方をしたならばどうなるでしょう．おそら

く日付の形式ごとに別々の名前で関数を用意し，ど

の形式で出力すべきか調べてから適切な関数を個別に

呼び出すようなコードになるでしょう．

それをこのコードではインスタンス化した後は，た

だ

$date->to_string;

とメソッドを呼び出すだけで済んでいます．条件フラ

グも，突然表れる謎の識別番号も必要ありません．サ

ブルーチンと処理が固定された関係にあるのとは異な

り，この例のようにメソッドに対応した処理が実行時

に決定されることを動的束縛と呼びます．

また例の中のスカラ変数$dataはDateStringクラス

のインスタンスであるときもあれば，DateStringRfc1123

クラスやDateStrinRfc850クラスのインスタンスである

ときもあります．このように同じメソッドを受け取る

ことができるオブジェクト同士を，置き換えて処理で

きるような特性をポリモフィズムと呼びます．

さらにPerlで動的束縛が行われる場面はインスタン

スとメソッドの関係に留まらず，クラスとコンストラ

クタやメソッド自身に対しても行われます．リスト7

は先ほどあげたコードと同じ結果を出力します．つま

りPerlはクラス名を元にしてそのクラスを直接インス

タンス化できるのです．たとえば以下のような記述で

もインスタンスを生成できます．

"DateString"->new;

また，メソッド名を文字列で指定することで呼び出

すメソッドを動的に指定することができます．

my $method = 'to_string';

$date->$method();

静的なプログラミング言語を使っていて感じる「こ

んな風に書けたら良いのになぁ」をPerlでは簡素に直

接記述することができます．

駆け足で説明を進めてしまいましたが，Perlのオブ

ジェクト指向機能は本稿末の参考文献で挙げた書籍な

どで詳細に解説されています．そこでもし不明な点が

あったならば，Perlのオブジェクト指向機能について

のメーリングリスト“Object-oriented with Perl”を

覗いて，主旨をご理解いただいた上で参加してみてく

ださい．何かお役にたてるかもしれませんので．

●リスト7 クラスを直接インスタンス化する例

#!/usr/bin/perl
use DateString;
use DateStringRfc1123;
use DateStringRfc850;

for my $class (qw(DateString DateStringRfc1123 DateStringRfc850)){
my $date = $class->new;
print $date->to_string, "\n";

}
__END__

その他の情報源

●リスト6 ポリモフィズムの例

#!/usr/bin/perl
use DateString;
use DateStringRfc1123;
use DateStringRfc850;

my @date;
push @date, DateString->new;
push @date, DateStringRfc1123->new;
push @date, DateStringRfc850->new;

foreach  my $date (@date) {
print $date->to_string, "\n";

}
__END__

Webアプリケーションで行われる処理は比較的定型

的な処理が多く，発生するコーディング上の問題や解

決すべき技術的なポイントも似通っています．ここで

は一般的なWebアプリケーションの開発中に表れるコ

ーディング上の問題を題材とし，デザインパターンを

参考にしてそれらを解決していってみましょう．

掲示板の参照と書き込みを行う場合を考えてみま

す．Webアプリケーションのほとんどは一定処理手順

を踏んで実行されるのですが，要求の種類を判定する

ための条件分岐が処理フェーズごとに表れてしまうコ

ードをよく見かけます（リスト8）．

このようにコードに条件分岐がちりばめられている

と理解の妨げになるばかりか，コードに変更を加える

たびに各々の条件分岐を意識しながら修正する必要が

生まれるため，修正に要するコストが高くなってしま

います．

ヒントを得るためにデザインパターンのパターンカ

タログを見ると，目的に「1つのオペレーションにア

ルゴリズムのスケルトンを定義しておき，その中のい

くつかのステップについては，サブクラスの定義に任

せることにする」を掲げるTemplate Methodパターン

が参考になりそうです．

ここで言うアルゴリズムは

¡要求の解析

¡必要な処理

¡コンテンツの生成

¡コンテンツの出力

などWebアプリケーションで行われる一連の流れを指

します．まずこの流れを定義するApplicationHandler

Template Methodパターン

●リスト8 Template Method適用前

#!/usr/bin/perl
use strict;
my @message = qw(

MESSAGE1
MESSAGE2
MESSAGE3

);

my $mode;
if ($ENV{QUERY_STRINGS} =̃  m{̂ mode=write}) {

$mode = 'write';
}
else {

$mode = 'show';
}

if ($mode eq 'write') {
if ($ENV{QUERY_STRINGS} =̃  m{̂ mode=write?message=(.*)$}) {

push @message, $1;
}

}

my $content;
if ($mode eq 'write') {

$content = join "\n", 'In Write Mode', @message;
} else {

$content = join "\n", 'In Show Mode', @message;
}

print 'Content-Type: text/html', "\n\n";
print $content;
__END__

●リスト9 ApplicationHandlerクラス

package ApplicationHandler;
use strict;

sub new
{

my $class = shift;
my $query = shift || $ENV{QUERY_STRINGS};
bless {

content => '',
query   => $query,
message => [qw(MessageA MessageB MessageC)],

}, $class;
}

sub compute
{

my $self = shift;
$self->do_task;
$self->do_create_content;
$self->do_return_content;

}

sub do_task {}

sub do_create_content
{

my $self = shift;
my $messages = $self->{message};
$self->{content} = join "\n", @$messages;

}

sub do_return_content 
{

my $self = shift;
print "Content-Type: text/plain\n";
print $self->{content};

}
1;
__END__

デザインパターンの実例

クラス（リスト9）を作成します．

ApplicationHandlerクラスではフェーズごとに割り

当てられたメソッドdo_task，do_create_content，

do_return_contentが定義されています．これらのメ

ソッドはcomputeメソッドから逐次呼び出されます

（図3）．

ApplicationHandlerクラスを使用するコード，リス

ト10を実行すると

Content-Type: text/plain

Hello Object World

という出力が得られます．

続いてメッセージを参照する場合の処理を定義する

ShowHandlerクラス（リスト11）を作成します．ま

た同様にメッセージを書き込む場合の処理を定義する

WriteHandlerクラス（リスト12）を作成します．

ShowHandlerとWriteHandlerクラスは，Application

Handlerクラスを継承します．ApplicationHandlerク

ラスで定義したcomputeメソッドはこれらのクラスに

も受け継がれるため，リスト13のように使用します．

このようにそれぞれの処理フェーズに散らばってい

た条件分岐は，生成するインスタンスを決定する部分

だけに集約され，見通しの良いコードになりました．

Template Methodパターンはオブジェクト指向プログ

ラミングの中でも基礎の基礎で，あらゆるところでそ

の実装を見ることができます．

先ほど挙げたコードの冒頭に，書き込みモードか読

み込みモードかを判断して必要なクラスのインスタン

スを生成する処理があります．要求を解釈して必要な

クラスのインスタンスを生成して返すという単純な処

理ですが，この判断のための知識をカプセル化するこ

●リスト13 ShowHandlerとWriteHandlerの使用

#!/usr/bin/perl

use ShowHandler;
use WriteHandler;
use strict;

my $application;
if  ($ENV{QUERY_STRINGS} =̃  /mode=write/) {

$application = WriteHandler->new;
} else {

$application = ShowHandler->new;
}
$application->compute;
__END__

●リスト10 ApplicationHandlerクラスの使用

#!/usr/bin/perl

use ApplicationHandler;
use strict;

my $application = ApplicationHandler->new;
$application->compute;
__END__

ApplicationHandler

compute()�
�
do_task()�
do_create_content()�
do_return_content()

$self->do_task();�
$self->do_create_content();�
$self->do_return_content();

図3／ApplicationHandlerクラスのメソッド呼び出し

●リスト11 ShowHandlerクラス

package ShowHandler;
use base 'ApplicationHandler';
use strict;

sub do_create_content
{

my $self = shift;
my $messages = $self->{message};
$self->{content} = join "\n", 'In Show Mode',  @$messages;

}
1;
__END__

●リスト12 WriteHandlerクラス

package WriteHandler;
use base 'ApplicationHandler';
use strict;

sub do_task
{

my $self = shift;
my $query_strings = $self->{query};
my $messages = $self->{message};

if ($query =̃  /new_message=(.*)$/) {
push @$message, $1;

}
}

sub do_create_content
{

my $self = shift;
my $messages = $self->{message};
$self->{content} = join "\n", 'In Write Mode', @$messages;

}
1;
__END__

とで，さらに見通しが良くなるはずです．仮にその判

断の処理が複雑だったとしても，カプセル化してあれ

ば利用する側は単にその機能を呼び出す手続きを知っ

ているだけでよいのです．

そのための実装はリスト14のようになり，この

ApplicationFactoryクラスはリスト15のようなコード

で使用します．

結果このコードは，

¡生成するクラスつまり実行する機能を決定する

ApplicationFactoryクラス

¡書き込みモードの動作を決定する

WriteHandlerクラス

¡読み出しモードの動作を決定する

ShowHandlerクラス

といった風に明確に役割が分担され，条件分岐が少

ないため理解しやすい構成になっています．

このようにインスタンスを生成するための手続きを

提供し，インスタンス化するクラスを決定するための

知識をカプセル化する構成をFactory Methodパター

ンと呼びます．単純にコンストラクタを使用してイン

スタンスを生成する場合と比べ，Factory Methodパ

ターンを利用した場合は，実際に生成するインスタン

スを変化させることができるために柔軟性が高まりま

す．

身近な例では，RDBMSのAPIであるDBIの

connect( )メソッドがFactory Methodパターンの実例

と言えます．

ちなみにPerlはクラスとコンストラクタの関係にも

動的束縛を使用することができるので，Factory

Methodパターンを利用することなく生成するインス

タンスを変化させることができます．このクラスとコ

ンストラクタの動的束縛を要する場合，インスタンス

化するクラスを文字列で指定します．

my $class_name = 'ShowHandler';

my $application = $class_name->new;

この記述で，設定ファイルによってインスタンス化

するクラスを指定するなどの高いカスタマイズ性を得

ることができます．ただよほど単純な場面を除き，素

直にFactory Methodパターンを使用したほうが好ま

しいでしょう．

●リスト14 ApplicationFactoryクラス

package ApplicationFactory;
use WriteHandler;
use ShowHandler;
use strict;

sub create
{

if  ($ENV{QUERY_STRINGS} =̃  /mode=write/) {
return WriteHandler->new;

} else {
return ShowHandler->new;

}
}
1;
__END__

●リスト15 ApplicationFactoryクラスの使用

#!/usr/bin/perl

use ApplicationFactory;
use strict;

my $application = ApplicationFactory->create;
$application->compute;
__END__

●リスト16 LoggingDecoratorクラス

package LoggingDecorator;
use strict;
our $AUTOLOAD;

sub new
{

my $class = shift;
my $component = shift;
bless { component => $component }, $class;

}

sub get_component
{

my $self = shift;
$self->{component};

}

sub AUTOLOAD
{

my $self = shift;
my @args = @_;
my $method = $AUTOLOAD;
$method =̃  s/.+://;
return if $method eq 'DESTROY';

printf STDERR "%s - %s->(%s)\n",
scalar localtime,
ref$self->get_component,$method,join(', ', @args);

$self->get_component->$method(@_);
}
1;
__END__

アプリケーションの挙動を追跡するために詳細なト

レースログを出力していると，デバッグの労力を大幅

に軽減できる場合があります．しかしログを出力する

ためのコードを各クラスに定義していると，本来の目

的ではない処理が混入して簡潔性が失われてしまいま

す．また継承しオーバライドしてログを出力するよう

拡張すると，それぞれのクラスごとにログ出力用のサ

ブクラスが必要になってしまい，クラスの数が膨れ上

がる恐れがあります．

デザインパターンのパターンカタログを見ると，目

的に「オブジェクトに責任を動的に追加する．サブク

ラス化よりも柔軟な機能拡張方法を提供する」と記

されたDecoratorパターンが参考にできそうです．

Decoratorパターンはオブジェクトを包み込み装飾

（デコレーション）することで，そのオブジェクトに新

しい機能を追加します．さらに包み込んだオブジェク

トに対して要求を転送するので，そのオブジェクトと

同じように振る舞うことができます．これらの拡張を

継承などの静的な手段ではなく，実行時に動的に機

能を拡張できることがポイントです．

ここでは，リスト16のように任意のオブジェクトに

トレースログ出力機能をデコレーションするLogging

Decoratorクラスを定義します．

LoggingDecoratorクラスは，任意のオブジェクト

にSTDERRへのトレースログ出力機能を追加します．

あるクラスで定義されていないメソッドを呼び出され

たとき，PerlはAUTOLOADメソッドを呼び出しま

す．呼び出そうとしていたメソッドは特種変数

$AUTOLOADを参照することで得ることができ，そ

れを元に本来その要求を受け取るべきオブジェクトに

要求を転送することができます（図4）．この方法は

オブジェクトを自動転送する手段として一般的なテク

ニックで，これらの機能をまとめたDelegateモジュー

ルなどがCPANに登録されています．

作成したLoggingDecoratorクラスの使用方法を見

てみます．LoggingDecoratorクラスは，コンストラ

クタの引数にトレースログ出力機能を追加したいオブ

ジェクトを指定します（リスト17）．

生成したインスタンス$logging_dbhはLogging

Decoratorクラスのインスタンスですが，受け取った

要求をすべてDBIのオブジェクトに転送するため，普

通のDBIのデータベースハンドラと同じように振る舞

います．そのためDBIのデータベースハンドラがもつ

機能をそのままに，すべての操作に対してメソッド

名・引数・タイムスタンプをSTDERRに出力する拡

張が加えられています．

Mon Nov 19 17:38:03 2001 - DBI::db->prepare(

DELETE FROM users WHERE age < ?

)

Mon Nov 19 17:38:04 2001 - DBI::db->commit()

Mon Nov 19 17:38:04 2001 - DBI::db->disconnect()

この例ではDBIのオブジェクトにロギング機能を追

●リスト17 LoggingDecoratorクラスの使用

#!/usr/bin/perl

use LoggingDecorator;
use DBI;
use strict;

my $dbh = DBI->connect('dbi:driver:dsn', '', '', {
RaiseError => 1, AutoCommit => 0
});

my $logging_dbh = LoggingDecorator->new($dbh);
eval {

my $state = $logging_dbh->prepare(q{
DELETE FROM users WHERE age < ?

});
$state->execute(18);
$logging_dbh->commit;

};
$logging_dbh->rollback if $@;
$logging_dbh->disconnect;
__END__

Client
LoggingDecorator

AUTOLOAD()

Component

method()

$a_component->method();

図4／要求の転送

加しましたが，他にもCGIモジュールやIO::Socketモ

ジュールなど，任意のオブジェクトにロギング機能を

追加することができます．またget_componentメソッ

ドで元のオブジェクトを取り出すこともできるので，任

意のタイミングでこの機能を取り外すことも可能です．

■ホワイトボックスな再利用
継承は，継承するクラスの内部構造に関する知識

をある程度要求するため，ホワイトボックスな再利用

と呼ばれます．つまり継承関係を持つことはそのクラ

スどうしが非常に強い結びつきを持つことであり，ベ

ースクラスの変更がサブクラスへ直接影響する危険性

があります．これはベースクラスの一部を変更すれば

すべてのサブクラスに対して反映されるメリットの裏

返しのデメリットと言えます．

■ブラックボックスな再利用
逆にDecoratorパターンなどのように委譲によって

オブジェクトの機能を拡張・変更してゆく方法は，ブ

ラックボックスな再利用と呼ばれます．ブラックボッ

クスな再利用はあるオブジェクトの振る舞いを拡張し

たい場合でも，そのオブジェクトの内部構造を知る必

要がないのです．また継承とは異なり，オブジェクト

の振る舞いや責任を実行時に決定・変更できる強力な

柔軟性も見のがすことはできません．

Webアプリケーションは，いかに資源の占有時間

を短くするか，試行錯誤の連続と言えます．特にフロ

ントエンドのApacheは利用できるスロットの数が限

られているため，長時間スロットを占有するようなア

プリケーションでは新たな要求をさばくためのスロッ

トが確保できず，全体的な待ち時間が増大してしまい

ます．

この状況を回避する手段として処理時間を短縮する

ことが基本として挙げられますが，これにも限界があ

ります．そのため，大量のトラフィックをさばくWeb

アプリケーションでは，処理の要求を一時的に保存し

て別のタイミングで処理を実行する非同期処理を行う

場合もあります．非同期処理を実現するための方法は

いくつか存在しますが，ここでは手軽なキューを利用

した非同期処理を実装します．

デザインパターンのパターンカタログを眺めると，

目的に「要求をオブジェクトとしてカプセル化するこ

とによって，異なる要求や，要求からなるキューやロ

グによりクライアントをパラメータ化する．また，取

り消し可能なオペレーションをサポートする」を掲げ

るCommandパターンがこの例に適切であることがわ

かります．Commandパターンを参考にしたキューの

実装QueueクラスとCommandクラスをhttp://perl.

Commandパターン

object object

object

Queue

object

図5／Queueクラスの動作

●リスト18 SendmailCommandクラス

package SendmailCommand;
use base 'Command';
use strict;

use constant MAIL_TEMPLATE => <<__MAIL_TEMPLATE__;
From: <%>
To: <%s>
Subject: %s

%s
__MAIL_TEMPLATE__

sub to
{

my $self = shift;
$self->{to} = shift if @_;
$self->{to};

}

sub execute
{

my $self = shift;
open MAIL, '| /usr/lib/sendmail -f info@example.jp';
printf MAIL MAIL_TEMPLATE,

'info@example.jp', $self->to, 'Demo Message', 'Hello World';
close MAIL;

}
1;
__END__

infoware.ne.jp/source/Queue-0.1.tar.gz からダウンロ

ードできるので今回はこれを題材に説明します．

Queueクラスはファイルを使用したキューの実装で，

任意のオブジェクトをキューに蓄積することができま

す．QueueクラスはStorableモジュールを使用してオ

ブジェクトを文字列に変換してファイルに格納し，取

り出すときはファイルの文字列を評価してオブジェク

トを復元します（図5）．

キューに投入する処理の定義は，Commandクラス

のサブクラスで定義します．メール送信処理を行う

SendmailCommandクラス（リスト18）を定義して

みます．

SendmailCommandクラスは，toメソッドで送信先

をセットした後にexecuteメソッドを実行するとメー

ルの送信を始めます．単体で利用することもできます

が，キューを使用する場合は送信先を設定した状態で

キューに投入します（リスト19）．

Webアプリケーション側ではこの操作で終了し，次

の要求に備えます．今度はこの要求をキューから取り

出して実行する別のプロセスを定義します（リスト20）．

このコードを起動するとキューからすべての

Commandオブジェクト順番に取り出しexecuteメソ

ッドを呼び出して逐一実行します．Perlのref関数で

取得したCommandオブジェクトのクラス名を元にク

ラスを動的にロードしています．

この例では，キューに投入され$commandへ取り

出されるオブジェクトはSendmailCommandクラスの

インスタンスだけですが，CreateNewUserAccount

Commandクラスなど他のオブジェクトがキューに投

入されていたとしても，executeメソッドを実装した

オブジェクトであれば問題なく実行することができま

す．ここでもポリモフィズムを有効活用することで条

件分岐が不要になることがわかります．

この例では投入済みのCommandオブジェクトをす

べてキューから取り出して実行していますが，サーバ

の負荷に応じて一定量以上同時に処理しないようにす

ることも可能です．さらにネットワークを経て共有で

きるキューを利用した場合は，Commandオブジェク

トの実行を複数のマシンで分散処理することも可能で

す．要求をするプロセスと要求を処理するプロセスを

分断することで，応答性の確保ばかりでなく処理能力

のスケーラビリティも同時に得ることができるのです．

設定ファイルを参照することで，アプリケーション

が出力するメッセージや使用するクラスを変更するこ

とが可能になり，柔軟なソフトウェアができあがりま

す．ところが設定ファイルを参照するために複数のク

ラスで個別に設定ファイルの読み込み処理を実装した

り，個々に設定ファイルを読み込むインスタンスを生

成していたりしては無駄が多くなります．もちろんグ

ローバル変数にそのインスタンスを保持して使い回す

方法も採れますが，横断的に利用されるグローバル変

数はコードの理解の妨げになり，保守上の障害になる

可能性も高いため，できるだけ利用は避けたいもので

す．

パターンカタログの目的に「あるクラスに対してイ

ンスタンスが1つしか存在しないことを保証し，それ

にアクセスするためのグローバルな方法を提供する」

を掲げるSingletonパターンが参考になりそうなので，

WEB +DB PRESS Vol.6● 103

が似合う人になろう　
特集2

Singletonパターン

●リスト19 SendmailCommandクラスの使用
（クライアント側）

#!/usr/bin/perl
use Queue;
use SendmailTask;
use strict;

my $command = SendmailTask->new;
$command->to('oyama@crayfish.co.jp');

my $queue = Queue->new('/path/to/directory');
$queue->add($command);
__END__

●リスト20 SendmailCommandクラスの使用
（サーバ側）

#!/usr/bin/perl
use Queue;
use strict;

my $queue = Queue->new('/path/to/directory');
while (my $command = $queue->get) {

eval {
my $class = ref $command;
eval "require $class";
$command->execute;

};
print STDERR $@ if $@;

}
__END__

このパターンを利用して無駄を省いてみましょう．

Singletonパターンを実装する方法はいくつかありま

すが，今回は最も手軽なクラス変数を利用した実装を

行います．すでに設定ファイルを読み込むConfクラ

スが存在するとします．そのクラスを基にインスタン

スを1つだけ生成する拡張をします（リスト21）．

作成したSingletonConfクラスには，新しくinstance

メソッドを追加しました．instanceメソッドは，

SingletonConfクラスのインスタンスがすでに生成さ

れているときはその生成済みのインスタンスを返し，

生成されていないときは新しくインスタンスを生成し

て返します．

これによって設定ファイルの情報を参照したい場所

ではリスト22のように，決まった文法でインスタンス

化して参照することができるようになります．Singleton

パターンを利用することで生成するConfクラスのイン

スタンスを1つだけに抑えることができたうえ，グロ

ーバル変数に頼る必要もなくなります．

デザインパターンは，ソフトウェア設計の定石を集

めたものだと冒頭で述べました．ソフトウェアの世界

も，たとえば囲碁の世界と同様で，定石を外した設計

は素人相手には通用するかもしれませんが，プロには

通用しません．

CPANなどのライブラリを活用することと同様に，

自分の取り組んでいる分野がすでにパターン化されて

いないか，まず調べてみてください．もしあなたがオ

ブジェクト指向システムに慣れ親しんだ人であれば，

パターンによって設計を洗練させ知識を整理し，人に

伝えることが容易になるはずです．もしあなたがオブ

ジェクト指向システムに触れて間もない人ならば，パ

ターンから良い設計や定石を学びとることができるは

ずです．

今回紹介したのはパターンのごく一部で，この他に

も有用なパターンがたくさん公開されています．ソフ

トウェアの開発に携わる人はぜひパターンに触れて，

見つけて，そして共有してみてください．それはあな

たとあなたたちのソフトウェア開発経験をさらに豊か

にしてくれるはずです．#

●リスト21 SingletonConfクラス

package SingletonConf;
use base 'Conf';
use strict;

our $Instance;
sub instance
{

my $class = shift;
return $Instance if defined $Instance;

$Instance = $class->new(@_);
return $Instance;

}

1;
__END__

●リスト22 SingletonConfクラスの使用

#!/usr/bin/perl
use strict;
use SingletonConf;

my $conf = SingletonConf->instance;
my $class_name = $conf->get('DEFAULT_LOGGING_CLASS');
my $log = $class_name->new;
__END__

定石という宝石

